

# NAG Fortran Library Routine Document

## F07NNF (ZSYSV)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07NNF (ZSYSV) computes the solution to a complex system of linear equations

$$AX = B,$$

where  $A$  is an  $n$  by  $n$  symmetric matrix and  $X$  and  $B$  are  $n$  by  $r$  matrices.

### 2 Specification

```

SUBROUTINE F07NNF (UPLO, N, NRHS, A, LDA, IPIV, B, LDB, WORK, LWORK,
1                INFO)
INTEGER          N, NRHS, LDA, IPIV(*), LDB, LWORK, INFO
complex*16     A(LDA,*), B(LDB,*), WORK(*)
CHARACTER*1     UPLO

```

The routine may be called by its LAPACK name `zsysv`.

### 3 Description

The diagonal pivoting method is used to factor  $A$  as  $A = UDU^T$ , if  $UPLO = 'U'$  or  $A = LDL^T$ , if  $UPLO = 'L'$ , where  $U$  (or  $L$ ) is a product of permutation and unit upper (lower) triangular matrices, and  $D$  is symmetric and block diagonal with 1 by 1 and 2 by 2 diagonal blocks. The factored form of  $A$  is then used to solve the system of equations  $AX = B$ .

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

### 5 Parameters

- 1: UPLO – CHARACTER\*1 *Input*  
*On entry:* if UPLO = 'U', the upper triangle of  $A$  is stored.  
 If UPLO = 'L', the lower triangle of  $A$  is stored.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the number of linear equations, i.e., the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .

- 3: NRHS – INTEGER *Input*  
*On entry:*  $r$ , the number of right-hand sides, i.e., the number of columns of the matrix  $B$ .  
*Constraint:*  $\text{NRHS} \geq 0$ .
- 4: A(LDA,\*) – **complex\*16** array *Input/Output*  
**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .  
*On entry:* the symmetric matrix  $A$ .  
 If UPLO = 'U', the leading  $n$  by  $n$  upper triangular part of A contains the upper triangular part of the matrix  $A$ , and the strictly lower triangular part of A is not referenced.  
 If UPLO = 'L', the leading  $n$  by  $n$  lower triangular part of A contains the lower triangular part of the matrix  $A$ , and the strictly upper triangular part of A is not referenced.  
*On exit:* if INFO = 0, the block diagonal matrix  $D$  and the multipliers used to obtain the factor  $U$  or  $L$  from the factorization  $A = UDU^T$  or  $A = LDL^T$  as computed by F07NRF (ZSYTRF).
- 5: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F07NNF (ZSYSV) is called.  
*Constraint:*  $\text{LDA} \geq \max(1, N)$ .
- 6: IPIV(\*) – INTEGER array *Output*  
**Note:** the dimension of the array IPIV must be at least  $\max(1, N)$ .  
*On exit:* details of the interchanges and the block structure of  $D$ , as determined by F07NRF (ZSYTRF). If  $\text{IPIV}(k) > 0$ , then rows and columns  $k$  and  $\text{IPIV}(k)$  were interchanged, and  $D(k, k)$  is a 1 by 1 diagonal block. If UPLO = 'U' and  $\text{IPIV}(k) = \text{IPIV}(k-1) < 0$ , then rows and columns  $k-1$  and  $-\text{IPIV}(k)$  were interchanged and  $D(k-1 : k, k-1 : k)$  is a 2 by 2 diagonal block. If UPLO = 'L' and  $\text{IPIV}(k) = \text{IPIV}(k+1) < 0$ , then rows and columns  $k+1$  and  $-\text{IPIV}(k)$  were interchanged and  $D(k : k+1, k : k+1)$  is a 2 by 2 diagonal block.
- 7: B(LDB,\*) – **complex\*16** array *Input/Output*  
**Note:** the second dimension of the array B must be at least  $\max(1, \text{NRHS})$ . To solve the equations  $Ax = b$ , where  $b$  is a single right-hand side, B may be supplied as a one-dimensional array with length  $\text{LDB} = \max(1, N)$ .  
*On entry:* the  $n$  by  $r$  right-hand side matrix  $B$ .  
*On exit:* if INFO = 0, the  $n$  by  $r$  solution matrix  $X$ .
- 8: LDB – INTEGER *Input*  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F07NNF (ZSYSV) is called.  
*Constraint:*  $\text{LDB} \geq \max(1, N)$ .
- 9: WORK(\*) – **complex\*16** array *Workspace*  
**Note:** the dimension of the array WORK must be at least  $\max(1, \text{LWORK})$ .  
*On exit:* if INFO = 0, WORK(1) returns the optimal LWORK.
- 10: LWORK – INTEGER *Input*  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F07NNF (ZSYSV) is called.  
 $\text{LWORK} \geq 1$ , and for best performance  $\text{LWORK} \geq \max(1, N \times nb)$ , where  $nb$  is the optimal blocksize for F07NRF (ZSYTRF).

If  $LWORK = -1$ , a workspace query is assumed; the routine only calculates the optimal size of the  $WORK$  array, returns this value as the first entry of the  $WORK$  array, and no error message related to  $LWORK$  is issued.

11: INFO – INTEGER

*Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If  $INFO = -i$ , the  $i$ th argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If  $INFO = i$ ,  $d_{ii}$  is exactly zero. The factorization has been completed, but the block diagonal matrix  $D$  is exactly singular, so the solution could not be computed.

## 7 Accuracy

The computed solution for a single right-hand side,  $\hat{x}$ , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and  $\epsilon$  is the *machine precision*. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where  $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$ , the condition number of  $A$  with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) and Chapter 11 of Higham (2002) for further details.

F07NPF (ZSYSVX) is a comprehensive LAPACK driver that returns forward and backward error bounds and an estimate of the condition number. Alternatively, F04DHF solves  $Ax = b$  and returns a forward error bound and condition estimate. F04DHF calls F07NNF (ZSYSV) to solve the equations.

## 8 Further Comments

The total number of floating point operations is approximately  $\frac{4}{3}n^3 + 8n^2r$ , where  $r$  is the number of right-hand sides.

The real analogue of this routine is F07MAF (DSYSV).

## 9 Example

To solve the equations

$$Ax = b,$$

where  $A$  is the complex symmetric matrix

$$A = \begin{pmatrix} -0.56 + 0.12i & -1.54 - 2.86i & 5.32 - 1.59i & 3.80 + 0.92i \\ -1.54 - 2.86i & -2.83 - 0.03i & -3.52 + 0.58i & -7.86 - 2.96i \\ 5.32 - 1.59i & -3.52 + 0.58i & 8.86 + 1.81i & 5.14 - 0.64i \\ 3.80 + 0.92i & -7.86 - 2.96i & 5.14 - 0.64i & -0.39 - 0.71i \end{pmatrix}$$

and

$$b = \begin{pmatrix} -6.43 + 19.24i \\ -0.49 - 1.47i \\ -48.18 + 66.00i \\ -55.64 + 41.22i \end{pmatrix}.$$

Details of the factorization of  $A$  are also output.

## 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F07NNF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NB, NMAX
PARAMETER       (NB=64,NMAX=8)
INTEGER          LDA, LWORK
PARAMETER       (LDA=NMAX,LWORK=NB*NMAX)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
*      .. Local Arrays ..
COMPLEX *16     A(LDA,NMAX), B(NMAX), WORK(LWORK)
INTEGER          IPIV(NMAX)
CHARACTER       CLABS(1), RLABS(1)
*      .. External Subroutines ..
EXTERNAL        X04DBF, ZSYSV
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07NNF Example Program Results'
WRITE (NOUT,*)
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read the upper triangular part of the matrix A from data file
*
READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
*
*      Read b from data file
*
READ (NIN,*) (B(I),I=1,N)
*
*      Solve the equations Ax = b for x
*
CALL ZSYSV('Upper',N,1,A,LDA,IPIV,B,N,WORK,LWORK,INFO)
*
IF (INFO.EQ.0) THEN
*
*      Print solution
*
WRITE (NOUT,*) 'Solution'
WRITE (NOUT,99999) (B(I),I=1,N)
*
*      Print details of factorization
*
WRITE (NOUT,*)
IFAIL = 0
CALL X04DBF('Upper', 'Non-unit diagonal', N, N, A, LDA,
+          'Bracketed', 'F7.4',
+          'Details of the factorization', 'Integer', RLABS,
+          'Integer', CLABS, 80, 0, IFAIL)
*
*      Print pivot indices
*

```

```

        WRITE (NOUT,*)
        WRITE (NOUT,*) 'Pivot indices'
        WRITE (NOUT,99998) (IPIV(I),I=1,N)
*
        ELSE
        WRITE (NOUT,99997) 'The diagonal block ', INFO,
+       ' of D is zero'
        END IF
        ELSE
        WRITE (NOUT,*) 'NMAX too small'
        END IF
        STOP
*
99999 FORMAT ((4X,4('(',F7.4,',',F7.4,') ')))
99998 FORMAT (1X,7I11)
99997 FORMAT (1X,A,I3,A)
        END

```

## 9.2 Program Data

F07NNF Example Program Data

```

      4                                     :Value of N
( -0.56,  0.12) ( -1.54, -2.86) (  5.32, -1.59) (  3.80,  0.92)
                ( -2.83, -0.03) ( -3.52,  0.58) ( -7.86, -2.96)
                (  8.86,  1.81) (  5.14, -0.64)
                ( -0.39, -0.71) :End matrix A
( -6.43, 19.24) ( -0.49, -1.47) (-48.18, 66.00) (-55.64, 41.22) :End vector b

```

## 9.3 Program Results

F07NNF Example Program Results

Solution

```
(-4.0000, 3.0000) ( 3.0000,-2.0000) (-2.0000, 5.0000) ( 1.0000,-1.0000)
```

Details of the factorization

```

      1           2           3           4
1 (-2.0954,-2.2011) (-0.1071,-0.3157) (-0.4823, 0.0150) ( 0.4426, 0.1936)
2                ( 4.4079, 5.3991) (-0.6078, 0.2811) ( 0.5279,-0.3715)
3                (-2.8300,-0.0300) (-7.8600,-2.9600)
4                (-0.3900,-0.7100)

```

Pivot indices

```

      1           2           -2           -2

```

---